
Problem description: 
 
This example represents the bending of a thick beam in 3D conditions. The beam is 
clamped at the left end and loaded by concentrated forces at the right end. The 
material is linear elastic. 
 
Two solutions are obtained. The first one without domain decomposition is used as a 
reference, while the second uses decomposition in three sub-domains, one of which is 
non-conforming with the other two. 
 
Numerical Solutions 
 
DOMA02 
 
This calculation uses no domain decomposition and is used to obtain a reference 
solution. It uses a regular mesh of 384 elements of type CUB8. 
 
The mesh generation file is: 
 
*%siz 50 
* 
OPTI ECHO 1 ELEM CUB8 DIME 3; 
opti titr 'DOMA - 02'; 
opti trac psc ftra 'doma02_mesh.ps'; 
* 
P1=0. 0. 0.; 
P2=0. 10. 0.; 
P3=0. 10. 10.; 
P4=0. 0. 10.; 
* 
P9=40. 5. 5.; 
P14=20. 5. 5.; 
* 
BASE = DALL (P1 D 4 P2) (P2 D 4 P3) (P3 D 4 P4) (P4 D 4 P1) PLAN; 
VECT=40. 0. 0.; 
* 
P1P = P1 PLUS VECT; 
P2P = P2 PLUS VECT; 
P3P = P3 PLUS VECT; 
P4P = P4 PLUS VECT; 
* 

BODY = VOLU BASE TRAN 24 VECT; 
ELIM 0.001 BODY (P9 ET P14 ET P1P ET P2P ET P3P et P4P); 
CLIM1 = P1P ET P2P ET P3P ET P4P; 
CLIM2 = BODY POIN DROI P1P P2P 0.001; 
CLIM2 = CLIM2 ET (BODY POIN DROI P2P P3P 0.001); 
CLIM2 = CLIM2 ET (BODY POIN DROI P3P P4P 0.001); 
CLIM2 = CLIM2 ET (BODY POIN DROI P4P P1P 0.001); 
SURF3 = BODY POIN PLAN P1P P2P P3P 0.001; 
TOUT = BODY ET CLIM1 ET CLIM2 ET SURF3; 
TASS TOUT; 
* 
OPTI SAUV FORMAT 'doma02.msh'; 
SAUV FORMAT TOUT; 
* 
OEIL1=-10000. -10000. 10000.; 
OEIL2=10000. -10000. 10000.; 
* 
TRAC OEIL2 BODY; 
TRAC CACH OEIL2 BODY; 
* 
fin; 

The input file is: 
 
DOMA - 02 
*----------------------------------------------------------------------- 
ECHO 
!conv win 
CAST tout  
*-----------------------------------------------------------Problem type 
TRID NONL 
*-----------------------------------------------------------Dimensioning 
DIME 
  PT3L 625 CUB8 384 
  MTPO 2 
  TABL 1 4 FORC 200 
TERM 
*---------------------------------------------------------------Geometry 
GEOM CUB8 body 
     TERM 
*--------------------------------------------------Geometric Complements 
COMP COUL vert LECT body TERM 
*----------------------------------------------------------Material data 
MATE  
   LINE RO 7800. YOUN 210E9 NU 0.  
        LECT body TERM 
*--------------------------------------------------------Applied "loads" 
CHAR 1 FACT 2  
   FORC 3 -1.E9 LECT clim1 TERM 
   FORC 3 -2.E9 LECT clim2 TERM 
   FORC 3  4.E9 LECT surf3 TERM 
   TABL 2 0.0 1.0 1.0 1.0 
*----------------------------------------------------Boundary conditions 
LINK COUP BLOQ 123 LECT base TERM 
*----------------------------------------------------------------Outputs 

ECRI DEPL TFRE 2.5E-2 
          POIN LECT p9 p14 TERM 
          NOEL 
     FICH ALIC TEMP FREQ 1 
          POIN LECT p9 p14 TERM  
          ELEM LECT 1 TERM 
     FICH ALIC TFRE 3.E-3 
*----------------------------------------------------------------Options 
OPTI CSTA 0.5 
     LOG 1 DPSD 
*--------------------------------------------------Transient calculation 
CALC TINI 0.0 TFIN 300.E-3 
*=========================================================POST-TREATMENT 
SUIT  
Post-treatment 
ECHO 
RESU ALIC TEMP GARD PSCR 
SORT GRAP 
AXTE 1E3 'Temps (ms)' 
*------------------------------------------------------Curve definitions 
COUR 1 'noeud P9'  DEPL COMP 3 NOEU LECT p9  TERM 
COUR 2 'noeud P14' DEPL COMP 3 NOEU LECT p14 TERM 
*------------------------------------------------------------------Plots 
TRAC 1 2 AXES 1 'Deplacement' yzer 
*--------------------------------------------------Results qualification 
QUAL DEPL COMP 3 LECT p9  TERM REFE 1.43266E+01 TOLE 1E-2 
     DEPL COMP 3 LECT p14 TERM REFE 4.82882E+00 TOLE 1E-2 
*======================================================================= 
FIN 
 

DOMA03 
 
This calculation uses decomposition into three sub-domains: 

• Domain 1 corresponds to ¼ of the beam and is coarser than the reference; 
• Domain 2 corresponds to ¼ of the beam, is also coarser than the reference and 

is conforming with domain 1. 
• Domain 3 corresponds to the remaining ½ of the structure. It has the same 

mesh as the reference and is therefore non-conforming with respect to both 
domain 1 and domain 2. 

 
Altogether, the mesh contains 264 elements of type CUB8 and 22 elements of type 
CL3D to define the interfaces between non-conforming sub-domains. 
 
The mesh generation file is: 
 
*%siz 50 
* 
OPTI ECHO 1 ELEM CUB8 DIME 3; 
opti titr 'DOMA - 03'; 
opti trac psc ftra 'doma03_mesh.ps'; 
* 
P1=0. 0. 0.; 
P2=0. 10. 0.; 
P3=0. 10. 10.; 
P4=0. 0. 10.; 
* 
P5=20. 0. 0.; 
P6=20. 10. 0.; 
P7=20. 10. 10.; 
P8=20. 0. 10.; 
* 
P10=0. 0. 5.; 
P11=0. 10. 5 .; 
P12=20. 0. 5.; 
P13=20. 10. 5 .; 
* 
P9=40. 5. 5.; 
P14=20. 5. 5.; 
* 
NY1=2; NZ1=1; 
NY2=2; NZ2=2; 
NY3=4; NZ3=2; 
NY32=4; NZ32=2*NZ3; 
NX1=12; NX2=12; 
* 
VECT=20. 0. 0.; 
* 
CLIM1=(P5 ET P6 ET P7 ET P8); 
CLIM1=CLIM1 PLUS VECT; 
CLIM2=(P5 D NY3 P6 D NZ32 P7 D NY3 P8 D NZ32 P5); 
CLIM2=CLIM2 PLUS VECT; 
* 
P1_P2=P1 D NY1 P2; 
P2_P11=P2 D NZ1 P11; 
P11_P10=P11 D NY1 P10; 
P10_P1=P10 D NZ1 P1; 
* 
SURF01=DALLER P1_P2 P2_P11 P11_P10 P10_P1; 
* 
P10_P11=P10 D NY2 P11; 
P11_P3=P11 D NZ2 P3; 
P3_P4=P3 D NY2 P4; 
P4_P10=P4 D NZ2 P10; 
* 
SURF02=DALLER P10_P11 P11_P3 P3_P4 P4_P10; 
* 
SURF0=SURF01 ET SURF02; 
* 
P6_P13=P6 D NZ3 P13; 
P13_P12=P13 D NY32 P12; 
P12_P13=P12 D NY32 P13; 
P12_P5=P12 D NZ3 P5; 
P13_P7=P13 D NZ3 P7; 
P8_P12=P8 D NZ3 P12; 

* 
P5_P6=P5 D NY32 P6; 
P6_P7=P6 D NZ32 P7; 
P7_P8=P7 D NY32 P8; 
P8_P5=P8 D NZ32 P5; 
* 
SURF2=DALLER P5_P6 P6_P7 P7_P8 P8_P5; 
SURF21=DALLER P5_P6 P6_P13 P13_P12 P12_P5; 
SURF22=DALLER P12_P13 P13_P7 P7_P8 P8_P12; 
* 
SURF11=SURF01 PLUS VECT; 
SURF12=SURF02 PLUS VECT; 
SURF3=SURF2 PLUS VECT; 
* 
P10_P11P=P10_P11 PLUS VECT; 
SURFC1=P10_P11 REGL NX1 P10_P11P; 
SURFC2=SURFC1 PLUS (0. 0. 0.); 
SURFC1=SURFC1 COUL ROUGE; 
SURFC2=SURFC2 COUL BLEU; 
* 
ELIM 0.001 SURF3 (CLIM1 ET CLIM2 ET P9); 
* 
ZONE1=SURF01 VOLU NX1 SURF11; 
ZONE2=SURF02 VOLU NX1 SURF12; 
ZONE3=SURF2 VOLU NX2 SURF3; 
* 
ELIM 0.001 ZONE1 (SURFC1 ET SURF0 ET SURF11); 
ELIM 0.001 ZONE1; 
ELIM 0.001 ZONE2 (SURFC2 ET SURF0 ET SURF12); 
ELIM 0.001 ZONE2; 
ELIM 0.001 ZONE3; 
ELIM 0.001 ZONE3 (SURF3 ET SURF21 ET SURF22 ET P14); 
* 
TOUT=ZONE1 ET ZONE2 ET ZONE3; 
* 
OPTI SAUV FORMAT 'doma03.msh'; 
SAUV FORMAT TOUT; 
* 
OEIL1=-10000. -10000. 10000.; 
OEIL2=10000. -10000. 10000.; 
* 
TRAC CLIM2; 
TRAC (CLIM1 ET CLIM2); 
TRAC (SURF3 ET CLIM1 ET CLIM2); 
TRAC SURF0; 
TRAC (SURF11 ET SURF12 ); 
TRAC SURF21; 
TRAC SURF22; 
TRAC (SURF21 ET SURF22 ); 
TRAC (SURF11 ET SURF12 ET SURF21 ET SURF22 ET SURFC1 ET SURFC2); 
TRAC CACH OEIL2 ZONE1; 
TRAC CACH OEIL2 ZONE2; 
TRAC CACH OEIL1 ZONE3; 
TRAC OEIL2 TOUT; 
TRAC CACH OEIL2 TOUT; 
* 
OPTI DONN 5; 
fin; 

 

The input file is: 
 
DOMA - 03 
*----------------------------------------------------------------------- 
ECHO 

!conv win 
CAST tout  
*-----------------------------------------------------------Problem type 

TRID NONL 
*-----------------------------------------------------------Dimensioning 
DIME 
  PT3L 1000 CUB8 1000 CL3D 1000 ZONE 2 
  MTPO 2 
  TABL 1 4 FORC 200 
TERM 
*---------------------------------------------------------------Geometry 
GEOM CUB8 tout 
     CL3D surf11 surf12 surf21 surf22 
     TERM 
*--------------------------------------------------Geometric Complements 
COMP COUL roug LECT zone1 surf11 TERM 
          vert LECT zone2 surf12 TERM 
          rose LECT zone3 surf21 surf22 TERM 
*----------------------------------------------------------Material data 
MATE  
   LINE RO 7800. YOUN 210E9 NU 0.  
        LECT zone1 zone2 zone3 TERM 
   FANT 0. 
        LECT surf11 surf12 surf21 surf22 TERM 
*--------------------------------------------------------Applied "loads" 
CHAR 1 FACT 2  
   FORC 3 -1.E9 LECT clim1 TERM 
   FORC 3 -2.E9 LECT clim2 TERM 
   FORC 3  4.E9 LECT surf3 TERM 
   TABL 2 0.0 1.0 1.0 1.0 
*----------------------------------------------------Boundary conditions 
LINK COUP BLOQ 123 LECT surf0 TERM 
*----------------------------------------------------------------Outputs 
ECRI DEPL TFRE 2.5E-2 
          POIN LECT p9 p14 TERM 
          NOEL 
     FICH ALIC TEMP FREQ 1 
          POIN LECT p9 p14 TERM  
          ELEM LECT 1 TERM 
     FICH ALIC TFRE 3.E-3 

*----------------------------------------------------------------Options 
OPTI CSTA 0.5 
     LOG 1 DPSD 
*---------------------------------------------------Domain decomposition 
STRU 3 
   DOMA LECT zone1 surf11        TERM 
   DOMA LECT zone2 surf12        TERM 
   DOMA LECT zone3 surf21 surf22 TERM 
INTE 3 
   COMP 
      DOMA 1 LECT surfc1 TERM 
      DOMA 2 LECT surfc2 TERM 
   MORT   
      DOMA 1 LECT surf11 TERM 
      DOMA 3 LECT surf21 TERM 
   MORT  
      DOMA 2 LECT surf12 TERM 
      DOMA 3 LECT surf22 TERM 
*--------------------------------------------------Transient calculation 
CALC TINI 0.0 TFIN 300.E-3 
*=========================================================POST-TREATMENT 
SUIT  
Post-treatment 
ECHO 
RESU ALIC TEMP GARD PSCR 
SORT GRAP 
AXTE 1E3 'Temps (ms)' 
*------------------------------------------------------Curve definitions 
COUR 1 'noeud P9'  DEPL COMP 3 NOEU LECT p9  TERM 
COUR 2 'noeud P14' DEPL COMP 3 NOEU LECT p14 TERM 
*------------------------------------------------------------------Plots 
TRAC 1 2 AXES 1 'Deplacement' yzer 
*--------------------------------------------------Results qualification 
QUAL DEPL COMP 3 LECT p9  TERM REFE 1.43266E+01 TOLE 1E-2 
     DEPL COMP 3 LECT p14 TERM REFE 4.82882E+00 TOLE 1E-2 
*======================================================================= 
FIN 

 
The tip displacement in the reference case is: 
 

 
 
The tip displacement in the case with domain decomposition is: 

The final displacement field in the reference case and in the case with domain 
decomposition are: 
 

 


